FaCiLe: A Functional Constraint Library
Release 1.0

N. Barnier P. Brisset

July 3, 2001

Preface

FaCiLe is a constraint programming library over integer finite domains written in OCaml[5]. It offers all usual constraints system facilities to create and handle finite domain variables, arithmetic expressions and constraints (possibly non-linear), built-in global constraints and search goals. FaCiLe allows to easily build user-defined constraints and goals (including recursive ones) from scratch or by combining simple primitives, making pervasive use of OCaml higher-order functionals to provide a simple and flexible user interface. As FaCiLe is an OCaml library and not "yet another language", the user benefits from type inference and strong typing discipline, high level of abstraction, modules and objects system, as well as native code compilation efficiency, garbage collection and replay debugger. All these features of OCaml (among many others) allow to prototype and experiment quickly: modeling, data processing and interface are implemented with the same powerful and efficient language.

This manual is not a document about constraint programming techniques but only a tool description. Users should be familiar with other constraint systems to easily apprehend FaCiLe through the reading of this manual. Beginners can easily find comprehensive information on the Web (e.g. `http://www.cs.unh.edu/ccc/archive/`). This manual is neither a course about functional programming and users should refer to the official Caml Web site at `http://caml.inria.fr/` and the OCaml manual [5] to obtain introductory (as well as advanced) material about the host language of FaCiLe. Hurried readers may also take a look at a short overview appearing in the ALP Newsletter [1].

Since OCaml forbids overloading, FaCiLe unusual looking operators might be a little disconcerting at first sight. Moreover, there is no implicit casting, so explicit conversions between variables (or integer) and arithmetic expressions are compulsory. These features lead to less concise expressions than with poorly typed languages, but then the user knows exactly which operation is executed by the system and cannot erratically mix values of different types. Furthermore, ML style higher-order functionals and powerful type system ease the design and processing of complex data structures without the need of syntactic sugar (iterators, mapping and folding are "native" in OCaml). So FaCiLe does not provide endlessly more and more ad hoc functions for each particular case to exhibit the smallest possible code for toy examples, but rather aims at providing simple building blocks and operators to combine them efficiently.

This manual is structured in two main parts:

1. The user's manual which starts with basic examples to give a taste of FaCiLe, then details the main concepts and eventually discusses more advanced subjects like the design of constraints and goals from scratch.

2. The reference manual which describes module by module all the functionalities of FaCiLe.

Numerous examples are provided all along the user's manual and more complete ones are available within the standard distribution in the `examples` directory, as well as a generic `Makefile`.

Eventually, we would like to thank our first thorough industrial beta-tester, Mattias Waldau, whose suggestions helped us to develop this release.

Good reading.

Foreword

Portability

FaCiLe requires only the OCaml system (release 3.01 or greater) and should work in any environment supporting this system. It is developed in a Linux environment on PC architecture but does not use any specificities of Unix. It should work on other operating systems (i.e. MS Windows, Mac OS, ...), provided that the installation process is customised to the environment.

FaCiLe Structure and Naming Conventions

The library is split into numerous modules and submodules. They are all included into the main module `Facile` which should be opened by any other modules using FaCiLe. All the modules are extensively described in part II of this documentation. We do not recommend to users to open modules in `Facile` but to use prefixed notations (e.g. function `post` of `Cstr` is written `Cstr.post`). Pseudo-module `Easy` is the exception and should be opened: it provides several aliases to most frequently used values (see 4.13).

To avoid interferences with other modules of the user, all the modules are aliased in the `Facile` module and implementation module files are all prefixed by `fcl_` (except of course `Facile` itself). For example, implementation of module `Gcc` is in file `fcl_gcc.ml` and alias

```
module Gcc = Fcl_gcc
```

is defined in `Facile` (`facile.ml`). This alias mechanism is entirely transparent to the user of FaCiLe except for the one interested by the implementation of the library. The only possible visibility of `Fcl_` prefix is given by the uncaught exceptions printer (e.g. `Fcl_stak.Fail` instead of `Stak.fail`).

The reference part of this documentation is automatically generated from module interfaces (`.mli`). Some available functions, types or modules are intentionally not documented or even hidden in `Facile` module. They are not intented to the casual user.

Values and types names try to benefit as much as possible from the modularity. For example, most of the types are named `t`: type of constraints is `Cstr.t`, type of domains is `Domain.t`... In the same way, printing functions are named `fprint`, constraints are named `cstr` (e.g. `Gcc.cstr`)...

Standard or *label* mode of the OCaml compiler (option `-labels`) may be used with the library. FaCiLe makes use of labels (labelled arguments) as less as possible; only optional arguments are labelled.

Compilation with FaCiLe

FaCiLe is provided as bytecode and native code[1] libraries.

Bytecode version is compiled with debugging information (`-g` option of `ocamlc`) and then can be used with the source-level replay debugger (`ocamldebug`). A lot of checks are done in this mode

[1] If supported by your architecture. See `http://caml.inria.fr/ocaml/portability.html`

and exceptions may be raised revealing bad usage of the system ("fatal" errors) or bugs in the system itself ("internal" errors). In the second case, diligent users should send a bug report to the developers.

In the native code version, these redundant checks are not done and this mode should be used only on well-tried code.

The `Makefile` in the `examples` directory of the distribution provides generic rules to compile with FaCiLe in both modes producing `.out` (bytecode) or `.opt` (native code) executables.

The library may also be used through linked toplevel produced with the following command (after installation):

```
ocamlmktop -o facile -I +facile facile.cma
```

This is the toplevel used in the inlined examples of this documentation and invoked with the command line:

```
./facile -I +facile
```

Availability

The FaCiLe distribution and documentation are available by anonymous FTP at:

`ftp://ftp.recherche.enac.fr/pub/facile`

There is also a web site for FaCiLe where general information can be found:

`http://www.recherche.enac.fr/opti/facile`

Questions, bug reports, ..., can be mailed to

`facile@recherche.enac.fr`

Installation

Installation of FaCiLe is described in the `README` file of the distribution. Below is a copy of the corresponding part:

```
INSTALLATION:

  All you need is the Objective Caml 3.01 (or greater) compiler and
standard Unix tools (make, ...).

  0) Configure the library. The single option of configuration is the
directory you want to put the library files in (facile.cma, facile.cmxa,
facile.a facile.cmi). Default is the subdirectory "facile" of the Ocaml
library directory (returned by "ocamlc -where").

./configure [--faciledir <target directory>]

  1) First compile the library with a simple

        make

  2) Then install the library with a (usually as root)
```

```
       make install

  3) Check the installation

       make check

     You should get a solution for the 8 queens problem.
```

Examples

The directory `examples` of the distribution contains some examples and a generic `Makefile` to compile files with FaCiLe.

Examples are taken from the classic litterature:

Queens Place queens on a chessboard

Golf Organize a golf tournament for 8 teams of 4 players

Magic To count and to be counted

Marriage Stabilize preferences among spouses

Tiles Tile a big square with small squares

Golomb Find optimal Golomb rulers

Coins Give back change for any amount

Prolog Use FaCiLe as a Prolog interpreter on a family tree problem

Seven_eleven My grocer's favorite arithmetic puzzle

Contents

Part I

User's Manual

Chapter 1

Getting Started

This first chapter introduces the overall framework of FaCiLe and gives a preliminary insight about its programming environment and functionalities.

OCaml code using FaCiLe facilities (file `csp.ml` in the following example) must be compiled with the library of object byte code `facile.cma` when batch compiling with `ocamlc`:

```
ocamlc -I +facile facile.cma csp.ml
```

and with the library of object code `facile.cmxa` for native compilation with `ocamlopt`:

```
ocamlopt -I +facile facile.cmxa csp.ml
```

provided that the standard installation of FaCiLe (and previously of the OCaml system of course) has been performed (see p. vi) and that the `facile.cm[x]a` files have been successfully created in the OCaml standard library directory. For larger programs, a generic Makefile can be found in directory `examples` (see p. vii).

It may however be convenient to use an OCaml custom toplevel to experiment toy examples or check small piece of serious (thus larger) code. A FaCiLe toplevel (i.e. in which `facile.cma` is pre-loaded) is easily built with the following command:

```
ocamlmktop -o facile -I +facile facile.cma
```

and invoked with:

```
./facile -I +facile
```

The two following sections give a quick overview of the main basic concepts of FaCiLe with the help of two very simple examples which are explained step by step.

1.1 Basics

We first give a slight taste of FaCiLe with the recurrent trivial problem of the Canadian flag: one has to repaint the Canadian flag (shown in figure 1.1) with its two original colors, red and white, so that two neighbouring areas don't have the same color and the maple leaf is red of course. The CSP model is desperately straightforward:

- one variable for each area l, c, r and m;
- all variables have the same domain [0..1], 0 being red and 1, white;
- one difference constraint for each adjacence $l \neq c$, $c \neq r$, $m \neq c$ and the maple leaf is forced to be red $m = 0$.

The following piece of code solves this problem:



Figure 1.1: The problem of the Canadian flag

maple.ml

```
open Facile
open Easy
let _ =
  (* Variables *)
  let red = 0 and white = 1 in
  let dom = Domain.create [red; white] in
  let l = Fd.create dom and c = Fd.create dom
  and r = Fd.create dom and m = Fd.create dom in
  (* Constraints *)
  Cstr.post (fd2e l <>~ fd2e c);
  Cstr.post (fd2e c <>~ fd2e r);
  Cstr.post (fd2e m <>~ fd2e c);
  Cstr.post (fd2e m =~ i2e red);
  (* Goal *)
  let var_list = [l;c;r;m] in
  let goal = Goals.List.labeling var_list in
  (* Search *)
  if Goals.solve goal then begin
    Printf.printf "l="; Fd.fprint stdout l;
    Printf.printf " c="; Fd.fprint stdout c;
    Printf.printf " r="; Fd.fprint stdout r;
    Printf.printf " m="; Fd.fprint stdout m;
    print_newline () end
  else
    prerr_endline "No solution"
```

```
unix% ocamlc -I +facile facile.cma maple.ml
unix% ./a.out
l=0 c=1 r=0 m=0
```

The new flag is therefore a faithful copy of the genuine one.

This small example introduces the following features of FaCiLe:

- The user interface to the library is provided by module `Facile` which gathers several specialized "submodules". We thus advise to open module `Facile` systematically to lighten FaCiLe functions calls. Most frequently used functions and submodules can then be directly accessed by opening module `Easy` (`open Easy`). Functions and modules names have been carefully chosen to avoid name clashes as much as possible with OCaml standard library when opening these two modules, but the "dot prefix" notation can still be used in case of fortuitous overlapping.

- The problem variables are created by a call to function `create` of module `Fd` (for **F**inite **d**omain, see 4.12) which takes a domain of type `Domain.t` as only argument. Domains are built and handled by functions of module `Domain` (see 4.4) like `Domain.create l` which creates a domain containing all integers of list `l`.
- `fd2e` and `i2e` constructs an expression respectively from a variable and an integer. More complex arithmetic expressions and constraints are built with infix operators (obtained by adding the suffix `~` to their integer counterparts) taking two expressions as arguments. Most usual arithmetic operators (not necessarily infix) are provided in module `Arith` (see 4.2).
- Function `post` from module `Cstr` adds a constraint to the constraint "store", which means that the constraint is taken into account and domain reduction is performed (as well as propagation on other variables).
- Here the search goal is a simple labeling of the list of all the problem variables `[l;c;r;m]` obtained by a call to function `labeling` of submodule `List` embedded in module `Goals` (see 4.7). The goal is therafter solved by a call to `solve` which returns `false` if a failure occurred and `true` otherwise.
- The solution is then printed using function `fprint` from module `Fd`, which prints a variable on an output channel, i.e. its domain if the variable is not instantiated and its value otherwise.

This piece of code illustrates a typical FaCiLe CSP solving with the following pervasive ordered structure:

1. data and variables declaration
2. constraints statements
3. search goal specification
4. goal solving, i.e. searching solution(s)

In the next section, a more sophisticated example will help to precisely describe how these features can be easily implemented with FaCiLe.

1.2 A Classic Example

We solve now the even more recurrent cryptarithmetic problem $SEND + MORE = MONEY$ (see figure 1.2) where each letter stands for a distinct digit with $M \neq 0$ and $S \neq 0$.

$$\begin{array}{cccccc} & S & E & N & D \\ + & M & O & R & E \\ \hline M & O & N & E & Y \end{array}$$

Figure 1.2: The $SEND + MORE = MONEY$ problem

We model this problem with one variable for each digit plus three auxilliary variables to carry over, and the subsequent four arithmetic constraints specifying the result of the addition as we would do by hand. The following program implement this model:

smm.ml

```
open Facile
open Easy
let _ =
  (* Variables *)
```

```
  let s = Fd.interval 0 9 and e = Fd.interval 0 9 and n = Fd.interval 0 9
  and d = Fd.interval 0 9 and m = Fd.interval 0 9 and o = Fd.interval 0 9
  and r = Fd.interval 0 9 and y = Fd.interval 0 9 in
  (* Constraints *)
  Cstr.post (fd2e m >~ i2e 0);
  Cstr.post (fd2e s >~ i2e 0);
  let digits = [|s;e;n;d;m;o;r;y|] in
  Cstr.post (Alldiff.cstr digits);
  let c = Fd.array 3 0 1 in (* Carry array *)
  let one x = fd2e x and ten x = i2e 10 *~ fd2e x in
  Cstr.post (             one d +~ one e =~ one y +~ ten c.(0));
  Cstr.post (one c.(0) +~ one n +~ one r =~ one e +~ ten c.(1));
  Cstr.post (one c.(1) +~ one e +~ one o =~ one n +~ ten c.(2));
  Cstr.post (one c.(2) +~ one s +~ one m =~ one o +~ ten m);
  (* Search goal solving *)
  if Goals.solve (Goals.Array.labeling digits) then begin
    let value = Fd.int_value in
    Printf.printf "  %d%d%d%d\n" (value s) (value e) (value n) (value d);
    Printf.printf "+ %d%d%d%d\n" (value m) (value o) (value r) (value e);
    Printf.printf "=%d%d%d%d%d\n" (value m) (value o) (value n) (value e) (value
 y)
  end else
    prerr_endline "No solution"
```

```
unix% ocamlc -I +facile facile.cma smm.ml
unix% ./a.out
  9567
+ 1085
=10652
```

We detail each step of the above example:

- Variables whose domains range integer intervals are created with function `Fd.interval inf sup` which creates a variable whose domain contains all integers between `inf` and `sup` (inclusive).
- Disequations $M \neq 0$ and $S \neq 0$ are then expressed by arithmetic inequality constraints and we assert that all digits must be distinct with the global `Alldiff.cstr` constraint which takes an array of variables as argument (see 4.1). FaCiLe provides some other global constraints as well, such as the global cardinality constraint (a.k.a. the "distribute" constraint) or the "sorting" constraint (see 4.6 and 4.10), embedded in separate module and called with function `cstr`.
- The three auxilliary carry variables are then created with `Fd.array n inf sup` which builds an array of `n` variables whose domains range the interval [`inf`..`sup`], and two auxilliary functions `one x` and `ten x` are defined which return an arithmetic expression being respectively equal to `x` and ten times `x` to lighten the main constraints statements.
- The equations reproducing the way we would compute the addition of figure 1.2 by hand are then straightforwardly stated and posted to the constraint store. The problem is finally solved as in the first example by a simple labeling of the decision variables, i.e. the "digits", using function `labeling` of module `Goals.Array` (which is the counterpart of `Goals.List` over arrays of variables). The solution is then printed with function `Fd.int_value` which returns the integer value of an instantiated variable (or raises an exception whenever it is still unbound).

We could of course have used a different but equivalent model constraining the addition to be exact without auxilliary carry variables:

```
...
let op1 =
  i2e 1000 *~ fd2e s +~ i2e 100 *~ fd2e e +~ i2e 10 *~ fd2e n +~ fd2e d
and op2 =
  i2e 1000 *~ fd2e m +~ i2e 100 *~ fd2e o +~ i2e 10 *~ fd2e r +~ fd2e e in
let result =
  i2e 10000 *~ fd2e m +~
  i2e 1000 *~ fd2e o +~ i2e 100 *~ fd2e n +~ i2e 10 *~ fd2e e +~ fd2e y in
Cstr.post (op1 +~ op2 =~ op3);
...
```

This alternative model would undoubtedly produce the same result.

The next chapter will explore in a more formal way how to handle the main concepts of FaCiLe introduced in the two previous examples.

Chapter 2

Building Blocks

2.1 Domains

Finite domains of integers are created, accessed and handled with functions of module `Domain` (described exhaustively in section 4.4). They are represented as functional objects of (abstract) type `Domain.t` and can therefore be shared. Domains are built with different functions according to the domain properties:

- `Domain.empty` is the empty domain;
- `Domain.create` is the most general constructor and builds a domain from a list of integers, possibly unsorted and with duplicates;
- `Domain.interval` is a shorthand when domains are continuous;
- `Domain.boolean` is a shorthand for `create [0;1]`;
- `Domain.int` is the largest (well, at least very large) domain.

Domains can be conveniently printed on an output channel with `Domain.fprint` and are displayed as lists of non-overlapping intervals and single integers `[`$\texttt{inf}_1$`-`$\texttt{sup}_1$`;`$\texttt{val}_2$`;`$\texttt{inf}_3$`-`$\texttt{sup}_3$`;...]` in increasing order:

```
#let discontinuous = Domain.create [4;7;2;4;-1;3];;
val discontinuous : Facile.Domain.t = <abstr>

#Domain.fprint stdout discontinuous;;
[-1;2-4;7]- : unit = ()

#let range = Domain.interval 4 12;;
val range : Facile.Domain.t = <abstr>

#Domain.fprint stdout range;;
[4-12]- : unit = ()
```

Various functions allow access to properties of domains like, among others (see 4.4), `Domain.is_empty`, `Domain.min`, `Domain.max` whose names are self-explanatory:

```
#Domain.is_empty range;;
- : bool = false

#Domain.max range;;
- : int = 12

#Domain.member 3 discontinuous;;
```

```
- : bool = true

#Domain.values range;;
- : int list = [4; 5; 6; 7; 8; 9; 10; 11; 12]
```

Operators are provided as well to handle domains and easily perform set operations like `Domain.intersection`, `Domain.union`, `Domain.difference` and domain reduction like `Domain.remove`, `Domain.remove_up`, `Domain.remove_low`, etc. (see 4.4):

```
#Domain.fprint stdout (Domain.intersection discontinuous range);;
[4;7]- : unit = ()

#Domain.fprint stdout (Domain.union discontinuous range);;
[-1;2-12]- : unit = ()

#Domain.fprint stdout (Domain.remove_up 3 discontinuous);;
[-1;2-3]- : unit = ()

#Domain.fprint stdout (Domain.remove_closed_inter 7 10 range);;
[4-6;11-12]- : unit = ()
```

2.2 Variables

FaCiLe variables are attributed objects[4] which maintain their current domain and can be backtracked during the execution of search goals.

Creation

FaCiLe finite domain constrained variables are build and handled by functions of module `Var.Fd` (described exhaustively in section 4.12). Variables are objects of type `Fd.t` created by a call to one of the following functions of module `Var.Fd`:

- `create d` takes a domain `d` as argument.
- `interval inf sup` yields a variable whose domain ranges the interval [`inf..sup`]. It is equivalent to `create (Domain.interval inf sup)`.
- `array n inf sup` creates an array of `n` "interval" variables. Equivalent to `Array.init n (fun _ -> Fd.interval inf sup)`.
- `int n` returns a variable already bound to `n`.

Note that the submodule `Fd` can be reached by opening module `Easy`; in all the toplevel examples, modules `Facile` and `Easy` are supposed open, therefore a function `f` of module `Fd` is called with `Fd.f` instead of `Facile.Var.Fd.f`.

The first three creation functions actually have an optional argument labelled `?name` which allows to associate a string identifier to a variable. The ubiquitous `fprint` function writes a variable on an output channel and uses this string name if provided or an internal identifier if not:

```
#let vd = Fd.create ~name:"vd" discontinuous;;
val vd : Facile.Var.Fd.t = <abstr>

#Fd.fprint stdout vd;;
vd[-1;2-4;7]- : unit = ()
```

Attribute

A FaCiLe variable can be regarded as either in one of the following two states:

- *uninstantiated* or *unbound*, such that an "attribute" containing the current domain (of size strictly greater than one) is attached to the variable;
- *instantiated* or *bound*, such that merely an integer is attached to the variable.

So an unbound variable is associated with an attribute of type `Var.Attr.t` holding its current domain, its string name, a unique integer identifier and various internal data irrelevant to the end-user. Functions to access attributes data are gathered in module `Var.Attr`:

- `dom` returns the current domain of an attribute;
- the mapping of `fprint`, `min`, `max`, `size`, `member` of module `Domain` applied on the embedded domain of an attribute (e.g. `min a` is equivalent to `Domain.min (dom a)`);
- `id` to get the identifier of an attribute;
- `constraints_number` returns the number of "active" constraints still attached to a variable.

Although variables are of abstract type `Fd.t`, function `Fd.value v` returns a concrete view of type `Var.concrete_fd = Unk of Attr.t | Val of int` [1] of a variable `v`, so that a control structure that depends on the instantiation of a variable will typically look like:

```
match Fd.value v with
    Val n -> f_bound n
  | Unk attr -> f_unbound attr
```

An alternative boolean function `Fd.is_var` returns the current state of a variable, sparing the "match" construct.

```
#let v1 = Fd.create (Domain.create [1]) (* equivalent to Fd.int 1 *);;
val v1 : Facile.Var.Fd.t = <abstr>

#Fd.is_var v1;;
- : bool = false

#Fd.fprint stdout v1;;
1- : unit = ()
```

Domain Reduction

Module `Fd` provides two functions to perform backtrackable domain reductions on variables, typically used within instantiation goals and filtering of user-defined constraints:

- `unify v n` instantiates variable `v` to integer `n` or fails whenever `n` does not belong to the domain of `v`. `unify` may be called on instantiated variables.

```
#let vr = Fd.interval 2 6;;
val vr : Facile.Var.Fd.t = <abstr>

#Fd.unify vr 7;;
Uncaught exception: Fcl_stak.Fail("Var.Fd.subst").

#Fd.unify vr 5;;
- : unit = ()
```

[1] Type `Var.concrete_fd` constructors `Unk` and `Val` stand respectively for "Unknown" (unbound) and "Value" (bound).

```
#Fd.fprint stdout vr;;
5- : unit = ()

#Fd.unify v1 2;;
Uncaught exception: Fcl_stak.Fail("Var.Fd.unify").

#Fd.unify v1 1;;
- : unit = ()
```

- `refine v dom` reduces the domain of `v` to `dom`. `dom` **must be included in the current domain of** `v` otherwise an assert failure is raised with the byte code library `facile.cma` or the system will be corrupted with the optimized native code library `facile.cmxa`.

```
#Fd.fprint stdout vd;;
vd[-1;2-4;7]- : unit = ()

#match Fd.value vd with
#    Val n -> () (* Do nothing *)
#  | Unk attr -> (* Remove every value > 2 *)
#      let new_dom = Domain.remove_up 2 (Var.Attr.dom attr) in
#      Fd.refine vd new_dom;;
- : unit = ()

#Fd.fprint stdout vd;;
vd[-1;2]- : unit = ()
```

Whenever the domain of a variable becomes empty, a failure occurs (see 2.5 for more explanations about failure):

```
#match Fd.value vd with
#    Val n -> () (* Do nothing *)
#  | Unk attr -> (* Remove every value < 4 *)
#      let new_dom = Domain.remove_low 4 (Var.Attr.dom attr) in
#      Fd.refine vd new_dom;;
Uncaught exception: Fcl_stak.Fail("Var.Fd.refine").
```

Access

Besides `Fd.value` and `Fd.is_var` which access the state of a variable, module `Fd` provides the mapping of module `Domain` functions like `Fd.size`, `Fd.min`, `Fd.max`, `Fd.values`, `Fd.iter` and `Fd.member`, and they return meaningful values whatever the state (bound or unbound) of the variable may be:

```
#let vr = Fd.interval 5 8;;
val vr : Facile.Var.Fd.t = <abstr>

#Fd.size vr;;
- : int = 4

#let v12 = Fd.int 12;;
val v12 : Facile.Var.Fd.t = <abstr>

#Fd.member v12 12;;
- : bool = true
```

Contrarily, function `Fd.id`, which returns the unique identifier associated with a variable, or function `Fd.name`, which returns its specified string name, only work if the variable is still uninstantiated, otherwise an exception is raised.

An order based on the integer identifiers is defined by function `Fd.compare`[2] as well as an equality function `Fd.equal`, observing the following two rules:

1. bound variables are smaller than unbound variables;
2. unbound variables are compared according to their identifiers.

```
#Fd.id vr;;
- : int = 6

#Fd.id v12;;
Uncaught exception: Failure "Fatal error: Fd.id: bound variable".

#Fd.compare v12 (Fd.int 11);;
- : int = 1

#Fd.compare vr v12;;
- : int = 1

#Fd.id vd;;
- : int = 4

#Fd.compare vd vr;;
- : int = -1
```

Eventually, function `Fd.int_value` returns the integer value of a bound variable. If the variable is not instantiated, an exception is raised.

```
#Fd.int_value (Fd.int 1);;
- : int = 1

#Fd.int_value (Fd.interval 0 1);;
Uncaught exception: Failure "Fatal error: Fd.int_value: unbound variable: ".
```

2.3 Arithmetic Expressions

Arithmetic expressions and constraints over finite domain variables are built with functions and operators of module `Arith` (see 4.2).

Creation and Access

Arithmetic expressions are objects of abstract type `Arith.t` which contain a representation of an arithmetic term over finite domain variables. An expression is *ground* when all the variables used to build it are bound; in such a state an expression can be "evaluated" with function `Arith.eval` which returns its unique integral value. A call to `Arith.eval` with an expression that is not ground raises the exception `Invalid_argument`. However, any expression can be printed on an output channel with function `Arith.fprint`.

A variable of type `Fd.t` or an OCaml integer of type `int` **are not** arithmetic expressions and cannot therefore be mixed up with the latter. "Conversion" functions are provided by module `Arith` to build an expression from variables and integers:

- `Arith.i2e n` returns an expression which evaluates to integer `n`;
- `Arith.fd2e v` returns an expression which evaluates to `n` when `v` is bound to `n`.

[2]Comparison functions return 0 if both arguments are equal, a positive integer if the first is greater than the second and a negative one otherwise (as specified in the OCaml standard library).

Handily enough, opening module `Easy` allows direct access to most useful functions and operators of module `Arith`, including `i2e` and `fd2e`:

```
#let v1 = Fd.interval 2 5;;
val v1 : Facile.Var.Fd.t = <abstr>

#let exp1 = fd2e v1;;
val exp1 : Facile.Arith.t = <abstr>

#Arith.fprint stdout exp1;;
_8[2-5]- : unit = ()

#Arith.eval exp1;;
Uncaught exception: Invalid_argument "Arith.eval: not ground".

#Fd.unify v1 4;;
- : unit = ()

#Arith.eval exp1;;
- : int = 4

#Arith.fprint stdout (i2e 2);;
2- : unit = ()
```

Maximal and minimal values of expressions can be accessed by functions `Arith.max_of_expr` and `Arith.min_of_expr`:

```
#let exp2 = fd2e (Fd.interval (-3) 12);;
val exp2 : Facile.Arith.t = <abstr>

#Arith.min_of_expr exp2;;
- : int = -3

#Arith.max_of_expr exp2;;
- : int = 12
```

Conversely, an arithmetic expression can be transformed into a variable thanks to function `Arith.e2fd` which creates a new variable constrained to be equal to its argument (see 2.4.2).

Operators

Module `Arith` provides classic linear and non-linear arithmetic operators to build complex expressions. Most frequently used ones can be directly accessed through the opening of module `Easy`, which considerably ligthen the writing of equations, especially for binary infix operators.

- `+~`, `-~`, `*~`, `/~`: addition, substraction, multiplication and division (the exception `Division_by_zero` is raised whenever its second argument evaluates to 0).
- `e **~ n` raises `e` to the `n`th power, where `n` is an integer.
- `x %~ y`: modulo. The exception `Division_by_zero` is raised whenever `y` evaluates to 0.
- `Arith.abs`: absolute value.

```
#let vx = Fd.interval ~name:"x" 3 6 and vy = Fd.interval ~name:"y" 4 12;;


#let exp1 = i2e 2 *~ fd2e vx -~ fd2e vy +~ i2e 3;;
val exp1 : Facile.Arith.t = <abstr>

#Arith.fprint stdout exp1;;
2*x[3-6]+-y[4-12]+3- : unit = ()
```

```
#Arith.min_of_expr exp1;;
- : int = -3

#Arith.max_of_expr exp1;;
- : int = 11
```

Global arithmetic operators working on array of expressions are provided as well:

- `Arith.sum exps` builds the sum of all the elements of the array of expressions `exps`.
- `Arith.scalprod ints exps` builds the scalar products of an array of integers by an array of expressions. `Arith.scalprod` raises `Invalid_argument` if the two arrays have not the same length.
- `Arith.prod exps` builds the product of all the elements of the array of expressions `exps`.

Their variable counterparts where the array of expressions is replaced by an array of variables are defined as well: `Arith.sum_fd`, `Arith.scalprod_fd`, `Arith.prod_fd`. Note that `Arith.sum_fd a`, for example, is simply defined as `Arith.sum (Array.map fd2e a)`.

```
#let size = 5;;
val size : int = 5

#let coefs = Array.init size (fun i -> i+1);;
val coefs : int array = [|1; 2; 3; 4; 5|]

#let vars = Fd.array size 0 9;;
val vars : Facile.Var.Fd.t array =
  [|<abstr>; <abstr>; <abstr>; <abstr>; <abstr>|]

#let pscal_exp = Arith.scalprod_fd coefs vars;;
val pscal_exp : Facile.Arith.t = <abstr>

#Arith.fprint stdout pscal_exp;;
0+_12[0-9]*1+_13[0-9]*2+_14[0-9]*3+_15[0-9]*4+_16[0-9]*5- : unit = ()

#Arith.min_of_expr pscal_exp;;
- : int = 0

#Arith.max_of_expr pscal_exp;;
- : int = 135
```

2.4 Constraints

2.4.1 Creation and Use

A constraint in FaCiLe is a value of type `Cstr.t`. It can be created by a built-in function (arithmetic, global constraints) or user-defined (see 3.3). A constraint must be *posted* with the function `Cstr.post` to be taken into account, i.e. added to the *constraint store*. The state of the system can then be accessed by a call to the function `Cstr.active_store` which returns the list of all constraints still "unsolved", i.e. not yet globally consistent.

When a constraint is posted, it is attached to the involved variables and activated: propagation occurs as soon as the constraint is posted. Consequently, if an inconsistency is detected prior to the search, i.e. before the call to `Goals.solve` (see 2.5), a `Stak.Fail` exception is raised. However, inconsistencies generally occur during the search so that failures are caught by the goal solving mechanism of FaCiLe which will backtrack until the last choice-point.

Constraints basically perform domain reductions on their involved variables, first when posted and then each time that a particular "event" occurs on their variables. An event corresponds to a

domain reduction on a variable: the minimal or maximal value has changed, the size of the domain has decreased or the variable has been bound. All these kinds reduction cause different events to trigger the "awakening" of the appropriate constraints. See 3.2.1 for a more precise description of this event-driven mechanism.

Constraints can also be printed on an output channel with function `Cstr.fprint` which usually yields useful information about the variables involved and/or the name of the constraint.

2.4.2 Arithmetic Constraints

The simplest and standard constraints are relations on arithmetic expressions (c.f. 2.3):

- equality `=~`
- strict and non-strict inequality `<~`, `>~`, `<=~`, `>=~`
- disequality `<>~`

FaCiLe provides them as infix operators suffixed with the `~` character, similarly to expression operators. These operators are declared in the `Easy` module and don't need module prefix notation whenever `Easy` is opened. The small example below uses the equality operator `=~` and points out the effect on the variables domains of posting the constraint `equation`:

```
#(* 0<=x<=10, 0<=y<=10, 0<=z<=10 *)
#let x = Fd.interval 0 10 and y = Fd.interval 0 10 and z = Fd.interval 0 10;;


#let equation = (* x*y - 2*z >= 90 *)
#fd2e x *~ fd2e y -~ i2e 2 *~ fd2e z >=~ i2e 90;;
val equation : Facile.Cstr.t = <abstr>

#(* before propagation has occurred *)
#Cstr.fprint stdout equation;;
 +2._19[0-10] -1._20[0-100] <= -90- : unit = ()

#Cstr.post equation;;
- : unit = ()

#(* after propagation has occurred *)
#Cstr.fprint stdout equation;;
 +2._19[0-5] -1._20[90-100] <= -90- : unit = ()
```

Notice that the output of the `Cstr.fprint` function does not look exactly like the stated inequation but gives a hint about how the two operands of the main sum are internally reduced into new single variables constrained to be equal to the latters. This mechanism is of course hidden to the user and is only unfolded when using the pretty-printer.

FaCiLe compiles and simplifies ("normalizes") arithmetic constraints as much as possible so that variables and integers may be scattered inside an expression with no loss of efficiency. Therefore the constraint `ineq1`:

```
#let x = Fd.interval (-2) 6 and y = Fd.interval 4 12;;

#let xe = fd2e x and ye = fd2e y;;


#let ineq1 = i2e 3 *~ ye +~ i2e 2 *~ xe *~ ye *~ i2e 5 *~ xe +~ ye >=~ i2e 4300;;
val ineq1 : Facile.Cstr.t = <abstr>

#Cstr.fprint stdout ineq1;;
 -4._24[4-12] -10._26[0-432] <= -4300- : unit = ()
```

which ensures $3y + (2xy \times 5x) + y \geq 4300$, i.e. $10x^2y + 4y \geq 4300$, is equivalent to `ineq2`:

```
#let ineq2 = i2e 10 *~ (xe **~ 2) *~ ye +~ i2e 4 *~ ye >=~ i2e 4300;;
val ineq2 : Facile.Cstr.t = <abstr>

#Cstr.fprint stdout ineq2;;
 -4._24[4-12] -10._31[0-432] <= -4300- : unit = ()
```

Once posted, `ineq1` or `ineq2` incidentally yield a single solution:

```
#Printf.printf "x=%a y=%a\n" Fd.fprint x Fd.fprint y;;
x=_23[-2-6] y=_24[4-12]
- : unit = ()

#Cstr.post ineq1;;
- : unit = ()

#Printf.printf "x=%a y=%a\n" Fd.fprint x Fd.fprint y;;
x=6 y=12
- : unit = ()
```

It is also worth mentioning that arithmetic constraints involving (large enough) sums of boolean variables are automatically detected by FaCiLe and handled internally by a specific efficient mechanism. The user may thus be willing to benefit from these features by choosing a suitable problem modeling.

Note on Overflows

Users should be carefull when expecting the arithmetic solver to compute bounds from variables with very large domain, that means with values close to `max_int` or `min_int` (depending on the system and architecture). Especially with exponentiation and multiplication, an integer overflow may occur which will yield an error message ("Fatal error: integer overflow") on `stderr` and an exception (`Assert_failure`) **if the program is compiled in byte code**. A spurious calculation (probably leading to a failure during propagation) will happen if it is compiled in native code. An unexpected behaviour when performing such operations in native code should thus always be checked against the safer byte code version.

2.4.3 Global Constraints

Beside arithmetic constraints, FaCiLe provides so-called "global constraints" which express a relation on a set of variables. They are defined in separate modules in which a function (and possibly several variants) usually named `cstr` yields the constraint; these functions takes an array of variables as their main argument.

The most famous one is probably the "all different" constraint which expresses that all the elements of an array of variables must take different values. This constraint is invoked by the function `Alldiff.cstr ?algo vars` where `vars` is an array of variables and `?algo` an optional argument (of type `Alldiff.algo`) that controls the efficiency of the constraint (see 4.1):

- `Lazy` waits for the instantiation of a variable and then removes the chosen value from the domains of the remaining variables;
- `Bin_matching evt` uses a more sophisticated algorithm (namely a "bin matching" [3]) which is called whenever the event `evt` (see 3.2.1) occurs on one of the variables to globally check the satisfiability of the constraint.

```
#let vars = Fd.array 5 0 4;;
val vars : Facile.Var.Fd.t array =
  [|<abstr>; <abstr>; <abstr>; <abstr>; <abstr>|]
```

```
#let ct = Alldiff.cstr vars;;
val ct : Facile.Cstr.t = <abstr>

#Fd.fprint_array stdout vars;;
[|_36[0-4]; _37[0-4]; _38[0-4]; _39[0-4]; _40[0-4]|]- : unit = ()

#Cstr.post ct; Fd.unify vars.(0) 3;;
- : unit = ()

#Fd.fprint_array stdout vars;;
[|3; _37[0-2;4]; _38[0-2;4]; _39[0-2;4]; _40[0-2;4]|]- : unit = ()
```

Module `FdArray` provides the "element" constraint named `FdArray.get` which allows to index an array of variables by a variable, and the `min` (resp. `max`) constraint which returns a variable constrained to be equal to the variable that will instantiate to the minimal (resp. maximal) value among the variables of an array:

```
#let vars = [|Fd.interval 7 12; Fd.interval 2 5; Fd.interval 4 8|];;
val vars : Facile.Var.Fd.t array = [|<abstr>; <abstr>; <abstr>|]

#let index = Fd.interval (-10) 10;;
val index : Facile.Var.Fd.t = <abstr>

#let vars_index = FdArray.get vars index;;
val vars_index : Facile.Var.Fd.t = <abstr>

#Fd.fprint stdout index;;
_50[0-2]- : unit = ()

#Fd.fprint stdout vars_index;;
_51[2-12]- : unit = ()

#let mini = FdArray.min vars;;
val mini : Facile.Var.Fd.t = <abstr>

#Fd.fprint stdout mini;;
_55[2-5]- : unit = ()
```

`FdArray.get` and `FdArray.min`, which produce a new variable (and thus hide an underlying constraint), also have their "constraint" counterpart `FdArray.get_cstr` and `FdArray.min_cstr` which take an extra variable as argument and return a constraint of type `Cstr.t` that must be posted to be effective: `FdArray.min_cstr vars mini` is therefore equivalent to the constraint:

```
fd2e (FdArray.min vars) =~ fd2e mini,
```

and `FdArray.get_cstr vars index v` to:

```
fd2e (FdArray.get vars index) =~ fd2e v.
```

More sophisticated global constraints are available as well as FaCiLe built-in constraints:

- the global cardinality constraint [6] (a.k.a. "distribute" constraint): `Gcc.cstr` (see 4.6);
- the sorting constraint [2]: `Sorting.cstr` (see 4.10).

2.4.4 Reification

FaCiLe constraints can be "reified" thanks to the `Reify` module and its function `Reify.boolean` (see 4.9) which takes an argument of type `Cstr.t` and returns a new boolean variable. This boolean variable is interpreted as the truth value of the relation expressed by the constraint and the following equivalences hold:

- the boolean variable is bound to 1 iff the constraint is satisfied, and the constraint is thereafter posted;
- the boolean variable is bound to 0 iff the constraint is violated, and the negation of the constraint is thereafter posted;

otherwise, i.e. it is not yet known if the constraint is satisfied or violated and the boolean variable is not instantiated, the reification of a constraint does not perform any domain reduction on the variables involved.

In the following example, the boolean variable `x_less_than_y` is constrained to the truth value of the inequation constraint $x < y$:

```
#let x = Fd.interval 3 6 and y = Fd.interval 5 8;;
val x : Facile.Var.Fd.t = <abstr>
val y : Facile.Var.Fd.t = <abstr>

#let x_less_than_y = Reify.boolean (fd2e x <~ fd2e y);;
val x_less_than_y : Facile.Var.Fd.t = <abstr>

#Fd.fprint stdout x_less_than_y;;
_60[0-1]- : unit = ()

#Cstr.post (fd2e y >=~ i2e 7);;
- : unit = ()

#Fd.fprint stdout x_less_than_y;;
1- : unit = ()

#Fd.fprint stdout (Reify.boolean (fd2e x =~ fd2e y));;
0- : unit = ()
```

When posted, the reification of a constraint calls the `check` function (see 3.3) of the constraint, which verifies whether it is satisfied or violated (without performing domain reduction). If it is violated, the negation of the constraint is posted with a call to another function of the constraint dedicated to reification, namely `not` (see 3.3). Both functions are always defined for all constraints but their default behaviour is merely exception raising (`Failure "Fatal error: ..."`) which means that the constraint is actually not reifiable - as specified in the documentation of the relevant constraints in the reference manual. Roughly, arithmetic constraints are reifiable (as well as the "interval" constraint of module `Interval`, see 4.8) while others (global ones) are not.

Reified constraint are by default woken up with the events triggering its standard awakening (i.e. as if it were directly posted) **and** those of its negation. This behaviour might possibly be too time costly (for some specific problem) and the call to `Reify.boolean` with its optional argument `?delay_on_negation` (see 4.9) set to `false` disables it, i.e. the events associated with the negation of the constraint are ignored.

Module `Reify` also provides standard logical (most of them infix) operators over constraints:

- `&&~~`, conjunction;
- `||~~`, disjunction;
- `=>~~`, implication;
- `<=>~~`, equivalence;
- `xor`[3], exclusive or;
- `not`[3], negation.

[3]Not infix.

These operators can be directly accessed through the opening of module `Easy`, except `Reify.not` (for obvious reasons) and `Reify.xor` (which are not infix). Note that, unlike `Reify.boolean`, these operators do not have a `?delay_on_negation` optional argument, so that the constraints they return will be woken by both the events of their arguments and those of the negations of their arguments.

These operators can be combined to yield complex logical operators. For example, the "exclusive or" may be redefined in the following way:

```
#let x = Fd.interval 3 5 and y = Fd.interval 5 7;;
val x : Facile.Var.Fd.t = <abstr>
val y : Facile.Var.Fd.t = <abstr>

#let xor ct1 ct2 = Reify.not (ct1 <=>~~ ct2) in
#let xor_cstr = xor (fd2e x =~ i2e 5) (fd2e y =~ i2e 5) in
#Cstr.post (xor_cstr);
#Cstr.post (fd2e x <=~ i2e 4);
#Printf.printf "x=%a y=%a\n" Fd.fprint x Fd.fprint y;;
x=_67[3-4] y=5
- : unit = ()
```

Furthermore, module `Arith` contains convenient shortcuts to reify its basic arithmetic constraints:

`=~~`, `<>~~`, `<=~~`, `>=~~`, `<~~`, `>~~`

These operators stand for the reification (and transformation into arithmetic expression) of their basic counterparts, i.e. they take two arithmetic expressions as operands and yield a new arithmetic expression being the boolean variable related to the truth value of the arithmetic constraint. `e1 =~~ e2` is therefore equivalent to

```
fd2e (Reify.boolean (e1 =~ e2))
```

These operators can also be directly accessed through the opening of module `Easy`. In the following example, the constraint stating that at least two of the three variables contained in array `vs` must be greater than 5 is expressed with the reified greater or equal `>=~~`:

```
#let vs = Fd.array 3 0 10;;
val vs : Facile.Var.Fd.t array = [|<abstr>; <abstr>; <abstr>|]

#Cstr.post (Arith.sum (Array.map (fun v -> fd2e v >~~ i2e 5) vs) >=~ i2e 2);
#Fd.fprint_array stdout vs;;
[|_76[0-10]; _77[0-10]; _78[0-10]|]- : unit = ()
```

If `vs.(1)` is forced to be less than 5, the two other variables become greater than 5:

```
#Cstr.post (fd2e vs.(1) <=~ i2e 5);
#Fd.fprint_array stdout vs;;
[|_76[6-10]; _77[0-5]; _78[6-10]|]- : unit = ()
```

2.5 Search

Most constraint models are not tight enough to yield directly a single solution, so that search (and/or optimization) is necessary to find appropriate ones. FaCiLe uses *goals* to search for solutions. All built-in goals and functions to create and combine goals are gathered in module `Goals` (see 4.7). This section only introduces "ready-to-use" goals intended to implement basic search strategies, but more experienced users shall refer to sections 3.1.2 and 3.4, where combining goals with iterators and building goals from scratch are explained.

FaCiLe's most standard labeling goals is `Goals.indomain` which instantiates non-deterministically a single variable by disjunctively trying each value still in its domain in increasing order. To be executed, a goal must then be passed as argument to function `Goals.solve` which returns `true` if the goal succeeds or `false` if it fails.

```
#let x = Fd.create (Domain.create [-4;2;12]);;
val x : Facile.Var.Fd.t = <abstr>

#Goals.solve (Goals.indomain x);;
- : bool = true

#Fd.fprint stdout x;;
-4- : unit = ()
```

So the first attempt to instantiate `x` (to -4) obviously succeeds.

The values of the domain of `x` can be enumerated with a slightly more sophisticated goal which fails just after `Goals.indomain`. Module `Goals` provides `Goals.fail`, which is a goal that always fails, and conjunction and disjunction operators, respectively `&&~` and `||~` (which can be directly accessed when module `Easy` is open), to combine simple goals. Hence such an enumeration goal would look like:

```
Goals.indomain x &&~ Goals.fail
```

But the result of such a goal will be failure and the state of the system (variable `x` not instantiated) will not be restored. A simple disjunction of this goal with the goal that always succeeds, `Goals.success`, yields the desirable behaviour:

```
(Goals.indomain x &&~ Goals.fail) ||~ Goals.success
```

In order to display the execution of this goal, a printing goal `gprint_fd` which prints a variable on the standard output (but will not be detailed in this section, see 3.4.1) can eventually be inserted (conjunctively) between `indomain` and `fail`:

```
#let x = Fd.create (Domain.create [-4;2;12]);;
val x : Facile.Var.Fd.t = <abstr>

#let goal = (Goals.indomain x &&~ gprint_fd x &&~ Goals.fail) ||~ Goals.success;;
val goal : Facile.Goals.t = <abstr>

#Goals.solve goal;;
-4 2 12 - : bool = true
```

Note that, unfortunately, **the logical operators do have the same priority**. Hence goals expressions must be carefully parenthesized to produce the expected result.

Module `Goals` also provides the function `Goals.instantiate` that allows to specify the ordering strategy of the labeling. `Goals.instantiate` takes as first argument a function to which is given the current domain of the variable (as single argument) and should return an integer candidate for instantiation. Labeling of variable `x` in decreasing order is then merely:

```
#let label_and_print labeling v =
#  (labeling v &&~ gprint_fd v &&~ Goals.fail) ||~ Goals.success;;
val label_and_print :
  (Facile.Var.Fd.t -> Facile.Goals.t) -> Facile.Var.Fd.t -> Facile.Goals.t =
  <fun>

#Goals.solve (label_and_print (Goals.instantiate Domain.max) x);;
12 2 -4 - : bool = true
```

Function `label_and_print` is defined here to lighten the writing of enumeration goals (it takes only the instantiation goal and the variable as arguments). In the example below, variable `x` is labelled in increasing order of the absolute value of its values. Function `Domain.choose` allows to only specify the relevant order:

```
#let goal =
#  label_and_print
#    (Goals.instantiate (Domain.choose (fun v1 v2 -> abs v1 < abs v2))) x;;
val goal : Facile.Goals.t = <abstr>

#Goals.solve goal;;
2 -4 12 - : bool = true
```

Beside non-deterministic instantiation, FaCiLe provides also `Goals.unify` to enforce the instantiation of a variable (which might be already bound) to a given integer value:

```
#Goals.solve (Goals.unify x 2);;
- : bool = true

#Fd.fprint stdout x;;
2- : unit = ()

#Goals.solve (Goals.unify x 12);;
- : bool = false

#Goals.solve (Goals.unify (Fd.int 0) 0);;
- : bool = true
```

Floundering If the search goal does not instantiate all the variables involved in the posted constraints, some of the constraints may still be unsolved when a solution is found, so that this solution may be incorrect. To be sure that all the constraints have been solved, the user can use the function `Cstr.active_store` and checks that the returned constraints list is empty. This checking may be done after the completion of the search, i.e. after `Goals.solve`, or better, embedded within the search goal. The latter allows to cleanly integrate this verification in optimization and "findall" goals. A "non-floundering check" goal could be implemented in the following way (function `Goals.atomic` used here to build a new atomic goal is explained in section 3.4.1):

```
#let check_floundering =
#  Goals.atomic
#    (fun () ->
#      if Cstr.active_store () <> [] then
#        failwith "Some constraints are still unsolved");;
val check_floundering : Facile.Goals.t = <abstr>
```

A simple conjunction with `check_floundering` at the end of the labeling goal will do the job. Information about the alive constraints may be extracted as well, thanks to module `Cstr` access functions (`id`, `name`, `fprint`).

Early Backtrack With FaCiLe as in Prolog systems, any dynamic modification performed within goals may be undone (backtracked) to restore the state of the system. However, no choicepoint is associated to the "root" of the constraint program, so that variables modifications occurring before the call to `Goals.solve` can never be undone. As the standard way of adding constraints with FaCiLe is to post them prior to the solving, i.e. statically outside goals, the domain reductions initially made by these constraints are not backtrackable.

2.6 Optimization

Classic Branch & Bound search is provided by the function `minimize` of module `Goals`. It allows to solve a specified goal (`g`) while minimizing a cost defined by a finite domain variable (`c`):

1. Goal `g` is solved and the cost must then be bound to a value *cc*, i.e. the current cost of the current solution
2. Backtracking is performed to restore the state of the system as before the execution of `g` and a new constraint stating `c` $< cc$ is added to the constraint store
3. The process loops until `goal` fails

The third argument of `Goals.minimize` is a function `solution : int -> unit` called each time a solution is found. The argument of `solution` is the current value of the cost `cc` which *must* be instantiated by `g`. This function is handy to store the last solution and cost in references, because `Goals.minimize` *always fails*, so that the decision and cost variables are restored as before its execution by `Goals.solve`.

The following example solves the minimization of $x^2 + y^2$ while $x + y = 10$:

```
#let x = Fd.interval 0 10 and y = Fd.interval 0 10 in
#Cstr.post (fd2e x +~ fd2e y =~ i2e 10);
#let c = Arith.e2fd (fd2e x **~ 2 +~ fd2e y **~ 2) in
#let store = ref None in
#let solution cc =
#  store := Some (cc, Fd.int_value x, Fd.int_value y);
#  Printf.printf "Found %d\n" cc in
#let g = Goals.minimize (Goals.indomain x &&~ Goals.indomain y) c solution in
#if Goals.solve (g ||~ Goals.success) then
#  match !store with
#    None -> Printf.printf "No solution\n"
#  | Some (best_c, best_x, best_y) ->
#      Printf.printf "Optimal solution: cost=%d x=%d y=%d\n" best_c best_x best_y;;
Found 100
Found 82
Found 68
Found 58
Found 52
Found 50
Optimal solution: cost=50 x=5 y=5
- : unit = ()
```

Additionally, `Goals.minimize` has two optional arguments:

- `?step`: the improvement between two consecutive solutions must be greater than `step`, i.e. the constraint posted each time a solution is found is `c` $\leq cc -$ `step`; `step` default value is obviously 1.
- `?mode`: may be either `Goals.Restart` or `Goals.Continue` (of type `bb_mode`); with mode `Restart`, the search restarts from the beginning at each step, i.e. the system backtracks until the very state prior to the execution of `minimize`, whereas with mode `Continue` the search simply carries on with an update of the cost constraint. Default mode is `Goals.Continue`.

Chapter 3

Advanced Usage

3.1 Search Control

3.1.1 Basic Mechanisms

FaCiLe implements a standard depth-first search with backtracking. OR control is handled with a stack (module `Stak`), while AND control is handled with continuations.

OR control can be modified with a cut à la Prolog: a level is associated to each choice-point (node in the search tree) and choice-points created since a specified level can be removed, i.e. *cut* (functions `Stak.level` and `Stak.cut`).

OR and AND controls are implemented by the `Goals.solve` function. AND is operationally mapped on the imperative sequence. OR is based on the exception mechanism: backtrack is caused by the exception `Stak.fail` which is raised by failing constraints. Note that this exception is caught and handled by the `Goals.solve` function only.

3.1.2 Combining Goals with Iterators

Functional programming allows the programmer to compose higher-order functions using *iterators*. An iterator is associated to a datatype and is the default control structure to process a value in the datatype. There is a strong isomorphism between the datatypes and the corresponding iterators and this isomorphism is a simple guideline to use them.

Imitating the iterators of the standard OCaml library, FaCiLe provides iterators for arrays and lists. While standard `Array` and `List` modules allows to construct sequences (with a '`;`') of imperative functions (type `'a -> unit`), `Goals.Array` and `Goals.List` modules of FaCiLe allows to construct conjunction (with a `&&~`) and disjunction (with a `||~`) of goals (type `Goals.t`).

The simplest iterator operates on integers and provides a standard *for-to* loop by applying a goal to consecutive integers:

```
Goals.forto 3 7 g = (g 3) &&~ (g 4) &&~ ... &&~ (g 7)
```

Of course, iterators may be composed, as is illustrated below, where the cartesian product $[1..3] \times [4..5]$ is deterministically enumerated:

```
#let enum_couples =
#  Goals.forto 1 3
#    (fun i ->
#      Goals.forto 4 5
#        (fun j ->
#          Goals.atomic (fun () -> Printf.printf "%d-%d\n" i j))) in
#Goals.solve enum_couples;;
1-4
```

```
1-5
2-4
2-5
3-4
3-5
- : bool = true
```

Function `Goals.atomic` (used in the previous example), which builds an "atomic" goal (i.e. a goal which returns nothing), is detailed in section 3.4.1.

Arrays: module `Goals.Array`

Standard Loop The polymorphic `Goals.Array.forall` function applies uniformally a goal to every element of an array, connecting them with a conjunction (`&&~`).

```
Goals.Array.forall g [|e1; e2; ...; en|] = (g e1) &&~ (g e2) &&~ ... &&~ (g en)
```

Labeling of an array of variables is the iteration of the instantiation of one variable (`Goals.indomain`):

```
#let labeling_array = Goals.Array.forall Goals.indomain;;
val labeling_array : Facile.Var.Fd.t array -> Facile.Goals.t = <fun>
```

A matrix is an array of arrays; following the isomorphism, labeling of a matrix must be simply a composition of the array iterator:

```
#let labeling_matrix = Goals.Array.forall labeling_array;;
val labeling_matrix : Facile.Var.Fd.t array array -> Facile.Goals.t = <fun>
```

Changing the Order An optional argument of `Goals.Array.forall`, labelled `?select`, gives the user the possibility to choose the order in which the elements are considered. `?select` is a function which is applied to the array by the iterator and which must return the index of one element on which the goal is applied. This function must raise the exception `Not_found` to stop the loop.

For example, if we want to apply the goal only on the unbound variables of an array, we may write:

```
#let first_unbound array =
#  let n = Array.length array in
#  let rec loop i = (* loop until free variable found *)
#    if i < n then
#      match Fd.value array.(i) with
#        Unk _ -> i
#      | Val _ -> loop (i+1)
#    else
#      raise Not_found in
#  loop 0;;
val first_unbound : Facile.Easy.Fd.t array -> int = <fun>

#let forall_unbounds = Goals.Array.forall ~select:first_unbound;;
val forall_unbounds :
  (Facile.Easy.Fd.t -> Facile.Goals.t) ->
  Facile.Easy.Fd.t array -> Facile.Goals.t = <fun>
```

Note that the function `forall` is polymorphic and can be used for an array of any type.

The function `Goals.Array.choose_index` facilitates the construction of heuristic functions that may be provided to the `forall ?select` argument. It constructs such a function from an ordering function on variable attributes (free variables are ignored). For example, the standard "min size" strategy will be implemented as follows:

```
#let min_size_order =
#  Goals.Array.choose_index (fun a1 a2 -> Var.Attr.size a1 < Var.Attr.size a2);;
val min_size_order : Facile.Var.Fd.t array -> int = <fun>

#let min_size_strategy = Goals.Array.forall ~select:min_size_order;;
val min_size_strategy :
  (Facile.Var.Fd.t -> Facile.Goals.t) ->
  Facile.Var.Fd.t array -> Facile.Goals.t = <fun>

#let min_size_labeling = min_size_strategy Goals.indomain;;
val min_size_labeling : Facile.Var.Fd.t array -> Facile.Goals.t = <fun>
```

Note that module `Goals.Array` also provides a disjunctive iterator, `exists`, which has the same profile than `forall`. Variants `Goals.Array.foralli` and `Goals.Array.existsi` allow to specify goals which take the index of the relevant variable as an extra argument (like the OCaml standard library iterator `Array.iteri`).

Lists: module `Goals.List`

FaCiLe `Goals.List` module provides similar iterators for lists except of course iterators which involve index of elements.

3.2 Constraints Control

Constraints may be seen operationally as "reactive objects". They are attached to variables, more precisely to events related to variable modifications. A constraint is mainly a function (the *update* function) which is called when the constraint is *woken*. The update function usually performs a propagation using the event, i.e. the modification of the domain of one variable, to process new domains for the other variables involved in the same constraint.

3.2.1 Events

An event (of type `Var.Attr.event`) is a modification of the the domain of a variable. FaCiLe currently provides four specific events:

- Modification of the domain (`on_refine`);
- Substitution of the variable, i.e. reduction of the domain to a singleton (`on_subst`);
- Modification of the minimum value of the domain (`on_min`);
- Modification of the maximum value of the domain (`on_max`).

Note that these events are not independant and constitute a lattice which top is `on_subst` and bottom is `on_refine`:

- `on_subst` implies all other events[1];
- `on_min` and `on_max` imply `on_refine`.

Constraints are attached to the variables through these events. In concrete terms, lists of constraints (one per event) are put in the attribute of the variable. Note that this attachement occurs only when the constraint is posted.

[1] It means that, e.g., the event `on_min` occurs even if a variable is instantiated to its minimum value.

3.2.2 Wakening, Queuing, Priorities

When an event occurs, related constraints are *woken* and put in a queue. The queue is processed after each sequence of waking. This processing is protected against reentrance. Constraints are considered one after the other and each update function is called to perform propagation. Propagation may fail by raising an exception or succeed. The propagation of one constraint is also protected against being woken again by itself. When a constraint is triggered, the update function does not know by which event, nor gets information about the variable responsible of it. A constraint is woken only once by two distinct events. Note also that the waking queue contains constraints and not variables.

FaCiLe implements three ordered queues and ensures that a constraint in a lower queue is not propagated before a constraint present in a higher queue. The queue is chosen according to the *priority* of a constraint (abstract type `Cstr.priority`). The priority is specified when the constraint is defined (see 3.3). It can be changed neither when the constraint is posted nor later. Priorities are defined in module `Cstr`: `immediate`, `normal` or `later`.

3.2.3 Constraint Store

FaCiLe handles the constraint store of all the *posted* and *active* constraints (a constraint becomes inactive if it is solved, i.e. if its update function returns true, see 3.3). For debugging purpose, this store can be consulted using the function `Cstr.active_store` and the returned constraints list may be processed using constraints (of type `Cstr.t`) access functions (`Cstr.id`, `Cstr.name` and `Cstr.fprint`).

3.3 User's Constraints

The `Cstr.create` function allows the user to build new constraints from scratch.

To define a new simple (unreifiable) constraint, very few arguments must be passed to the `create` function as numbers of them are optional (thus labelled) and have default values. Merely the two following arguments are actually needed to build a new constraint:

- `update` should perform propagation (domains reduction) and return true iff the constraint is consistent;
- `delay` specifies on which events the `update` function will be called. As shown in the following example, `delay` takes the constraint itself as its only argument and passes it to one or several calls to `Var.delay`.

However we recommend to name a new constraint and precise its printing facility, which may obviously help debugging, by specifying the following two optional arguments:

- `?name` should be a relevant string describing the purpose of the constraint;
- `?fprint` to print more accurate information on the constraint state (variables domains, maintained data structures value, ...).

To define a reifiable constraint, two additional optional arguments must also be specified:

- `?check` should return true if the constraint is entailed, false if its negation is entailed and raise the exception `DontKnow` otherwise. `check` is called when the constraint is reified and should therefore not perform any domain modification.
- `?not` should return the negation of the constraint (which is a constraint itself). It is called when the negation of a reified constraint is entailed, and to access the waking conditions of the negation of a constraint when its reification is posted (and the optional argument `?delay_on_negation` of `Reify.boolean` is set to `true` - which is its default value). Logical operators of module `Reify` also call the `?not` function for the same purpose, but they systematically do it (see 2.4.4).

Finally two other optional arguments may be specified:

- `?priority` should be passed to the `create` function to precise the priority of the new constraint in the constraints queue. Constraints with lower priority are waken only when there is no more constraint of higher priority in the waking queue. Time costly constraints should get a `later` while quick elementary constraints should be `immediate`, and standard constraints `normal` (default value).
- `?init` is executed as soon as the `post` function is called on the constraint to perform initialization of inner data structures needed by `update` (thus not called when dealing with a reified constraint).

The example below defines a new constraint stating that variable `x` should be different from variable `y`:

diff.ml

```
open Facile
open Easy

let cstr x y =
  let name = "different" in
  let fprint c =
    Printf.fprintf c "%s: %a <> %a\n" name Fd.fprint x Fd.fprint y
  and delay ct =
    Var.delay [Var.Attr.on_subst] x ct;
    Var.delay [Var.Attr.on_subst] y ct
  and update () =
    (* Domain reduction is performed only when x or y is instantiated *)
    match (Fd.value x, Fd.value y) with
      (Val a, Val b) -> a <> b || Stak.fail name
      (* If one of the two variables is instantiated, its value is
         removed in the domain of the other variable *)
    | (Val a, Unk attr_y) ->
        let new_domy = Domain.remove a (Var.Attr.dom attr_y) in
        Fd.refine y new_domy;
        true (* Constraint is solved *)
    | (Unk attr_x, Val b) ->
        let new_domx = Domain.remove b (Var.Attr.dom attr_x) in
        Fd.refine x new_domx;
        true (* Constraint is solved *)
    | _ -> false  (* Constraint is not solved *)
  and check () =
    match (Fd.value x, Fd.value y) with
      (Val a, Val b) -> a <> b
    | (Val a, Unk attr_y) when not (Var.Attr.member attr_y a) -> true
    | (Unk attr_x, Val b) when not (Var.Attr.member attr_x b) -> true
    | (Unk attr_x, Unk attr_y) when
        let dom_x = Var.Attr.dom attr_x and dom_y = Var.Attr.dom attr_y in
        Domain.is_empty (Domain.intersection dom_x dom_y) -> true
    | _ -> raise Cstr.DontKnow
  and not () =
    fd2e x =~ fd2e y in
    (* Creation of the constraint. *)
  Cstr.create ~name ~fprint ~check ~not update delay;;
```

Let's compile the file

```
unix% ocamlc -c -I +facile diff.ml
```

and use the produced object:

```
##load "diff.cmo";;

#let x = Fd.interval 1 2 and y = Fd.interval 2 3;;
val x : Facile.Easy.Fd.t = <abstr>
val y : Facile.Easy.Fd.t = <abstr>

#let diseq = Diff.cstr x y;;
val diseq : Facile.Cstr.t = <abstr>

#Cstr.post diseq;;
- : unit = ()

#let goal =
#  Goals.indomain x &&~ Goals.indomain y
#  &&~ Goals.atomic (fun () -> Cstr.fprint stdout diseq)
#  &&~ Goals.fail in
#while (Goals.solve  goal) do () done;;
different: 1 <> 2
different: 1 <> 3
different: 2 <> 3
- : unit = ()
```

Another example to test the reification function `check`:

```
#let x = Fd.interval 1 2 and y = Fd.interval 3 4;;
val x : Facile.Easy.Fd.t = <abstr>
val y : Facile.Easy.Fd.t = <abstr>

#let reified_diseq = Reify.boolean (Diff.cstr x y);;
val reified_diseq : Facile.Var.Fd.t = <abstr>

#Fd.fprint stdout reified_diseq;;
1- : unit = ()
```

3.4 User's Goals

3.4.1 Atomic Goal: `Goals.atomic`

The simplest way to create a deterministic atomic goal is to use the `Goals.atomic` function which "goalifies" any unit function (i.e. of type `unit -> unit`).

Let's write the goal which writes a variable on the standard output:

```
#let gprint_fd x = Goals.atomic (fun () -> Printf.printf "%a\n" Fd.fprint x);;
val gprint_fd : Facile.Easy.Fd.t -> Facile.Goals.t = <fun>
```

To instantiate a variable inside a goal, we may write the following definition:

```
#let unify_goal x v = Goals.atomic (fun () -> Fd.unify x v);;
val unify_goal : Facile.Easy.Fd.t -> int -> Facile.Goals.t = <fun>

#let v = Fd.interval 0 3 in
#if Goals.solve (unify_goal v 2) then Fd.fprint stdout v;;
2- : unit = ()
```

Note that this goal is the built-in goal `Goals.unify`.

This goal creation can be used to pack any side effect function:

```
#let gprint_int x = Goals.atomic (fun () -> print_int x);;
val gprint_int : int -> Facile.Goals.t = <fun>

#Goals.solve (Goals.forto 0 5 gprint_int);;
012345- : bool = true
```

The FaCiLe implementation of the classic "findall" of Prolog also illustrates the use of `Goals.atomic` to perform side effects: in this case to store all the solutions found in a list. The function `findall` in this example takes a "functional goal" `g` as argument which itself takes the very variable `x` from which we want to find all the possible values for which `g` succeeds; it could correspond to the Prolog term:

```
findall(X, g(X), Sol)
```

```
#let findall g x =
#  let sol = ref [] in
#  let store = Goals.atomic (fun () -> sol := Fd.int_value x :: !sol) in
#  let goal = g x &&~ store &&~ Goals.fail in
#  ignore (Goals.solve goal);
#  !sol;;
val findall :
  (Facile.Easy.Fd.t -> Facile.Goals.t) -> Facile.Easy.Fd.t -> int list =
  <fun>
```

We first declare a reference `sol` on an empty list to store all the solutions. Then the simple goal `store` is defined to push any new solution on the head of `sol` – note that we here use `Fd.int_value v` (see 4.12) for conciseness but it is quite unsafe unless we are sure that `v` is bound. The main goal is the conjunction of `g`, `store` and a failure. This goal obviously always fails, so we "ignore" the boolean returned by `Goals.solve`, and the solutions list is eventually returned.

The main point when creating goals is to precisely distinguish the time of *creation* of the goal from the time of its *execution*. For example, the following goal does not produce what might be expected:

```
#let wrong_min_or_max var =
#  let min = Fd.min var and max = Fd.max var in
#  (Goals.unify var min ||~ Goals.unify var max);;
val wrong_min_or_max : Facile.Easy.Fd.t -> Facile.Goals.t = <fun>
```

The `min` and `max` of the variable `var` are processed when the goal is created and may be different from the min and max of the variable when the goal will be called. To fix the problem, `min` and `max` must be computed within the goal. Then the latter must return the disjunction, which cannot be done with a simple call to `Goals.atomic`; function `Goals.create` (described in the next section) must be used instead.

3.4.2 Arbitrary Goal: `Goals.create`

The function `Goals.atomic` does not allow to construct goals which themselves construct new goals (similar to Prolog clauses). The `Goals.create` function "goalifies" a function which must return another goal, possibly `Goals.success` to terminate.

Let's write the goal which tries to instantiate a variable to its minimum value or to its maximum:

```
#let min_or_max v =
#  Goals.create
#    (fun () ->
#      let min = Fd.min v and max = Fd.max v in
```

```
#      Goals.unify v min ||~ Goals.unify v max)
#    ();;
val min_or_max : Facile.Easy.Fd.t -> Facile.Goals.t = <fun>
```

The other difference between `Goals.create` and `Goals.atomic` is the argument of the goalified function which may be of any type (`'a`) and which must be passed as the second argument to `Goals.create`. In the previous example, we use `()`.

`Goals.create` allows the user to define recursive goals by a mapping on a recursive function. In the next example, we iterate a goal non-deterministically on a list. Note that this goal is equivalent to the built-in goal `Goals.List.exists`.

```
#let rec iter_disj fgoal list =
#  Goals.create
#    (function
#         [] -> Goals.success
#       | x::xs -> fgoal x ||~ iter_disj fgoal xs)
#    list;;
val iter_disj : ('a -> Facile.Goals.t) -> 'a list -> Facile.Goals.t = <fun>

#let gprint_int x = Goals.atomic (fun () -> print_int x);;
val gprint_int : int -> Facile.Goals.t = <fun>

#let gprint_list = iter_disj gprint_int;;
val gprint_list : int list -> Facile.Goals.t = <fun>

#if Goals.solve (gprint_list [1;7;2;9] &&~ Goals.fail ||~ Goals.success) then
#  print_newline ();;
1729
- : unit = ()
```

3.4.3 Recursive Goals: `Goals.create_rec`

FaCiLe provides also a constructor for intrinsically recursive goals. Expression `Goals.create_rec f` is similar to `Goals.create f` except that the argument of the function `f` is the created goal itself.

The simplest example using this feature is the classic `repeat` predicate of Prolog implementing a non-deterministic loop:

```
#let repeat = Goals.create_rec (fun self -> Goals.success ||~ self);;
val repeat : Facile.Goals.t = <abstr>
```

The goalified function simply returned the disjunction of a success and itself.

The `Goals.indomain` function which non-deterministically instantiates a variable is written using `Goals.create_rec`:

```
#let indomain var =
#  Goals.create_rec ~name:"indomain"
#    (fun self ->
#      match Fd.value var with
#        Val _ -> Goals.success
#      | Unk attr ->
#         let dom = Var.Attr.dom attr in
#         let remove_min =
#           Goals.atomic (fun () -> Fd.refine var (Domain.remove_min dom))
#         and min = Domain.min dom in
#         Goals.unify var min ||~ (remove_min &&~ self));;
val indomain : Facile.Easy.Fd.t -> Facile.Goals.t = <fun>
```

The goal first checks if the variable is already bound and does nothing in this case. If it is an unknown, it returns a goal trying to instantiate the variable to its minimum or to remove it before continuing with the remaining domain.

Part II

Reference Manual

Chapter 4

Modules

The following sections are extracted from `.mli` module interfaces of FaCiLe and sorted by alphabetic order. A comprehensive index of these sections can be found at the end of this document.

4.1 Module `Alldiff`: the "All Different" Constraint

```
type algo = Lazy | Bin_matching of Var.Attr.event
```

```
val cstr : ?algo:algo -> Var.Fd.t array -> Cstr.t
```

> `alldiff (?algo:Lazy) vars` States that variables of `vars` are different from each other. The optional argument `algo` specifies the level of propagation. `Lazy`: waits for instantiation and removes from other domains. `Bin_matching c`: waits for event `c` (e.g. `Var.Attr.on_refine`) and uses a bin matching algorithm to ensure the constraint is consistent. `algo` default value is `Lazy`. Not reifiable.

4.2 Module `Arith`: Arithmetic Expressions over Variables of Type `Var.Fd.t`

Overview

This module provides functions and operators to build arithmetic expressions and state (in/dis)equation constraints over them.

Basic

```
type t
```

> Type of arithmetic expressions over variables of type `Var.Fd.t` and integers.

```
val fprint : out_channel -> t -> unit
```

> `fprint chan e` prints expression `e` on channel `chan`.

```
val eval : t -> int
```

> `eval e` returns the integer numerical value of a fully instantiated expression `e`. Raises `Invalid_argument` if `e` is not instantiated.

```
val min_of_expr : t -> int
```

```
val max_of_expr : t -> int
```

`min_of_expr e` (resp. `max_of_expr e`) returns the minimal (resp. maximal) possible value of expression `e`.

Conversion

```
val i2e : int -> t
```

`i2e n` returns an expression which evaluates to `n`.

```
val fd2e : Var.Fd.t -> t
```

`fd2e v` returns an expression which evaluates to `n` when `v` is instantiated and `Var.Fd.value v` evaluates to `Val n`.

```
val e2fd : t -> Var.Fd.t
```

`e2fd e` creates and returns a new variable `v` and posts the constraint `fd2e v =~ e`.

Construction of Arithmetic Expressions

The arithmetic operators check whether any integer overflow (i.e. the result of an arithmetic operation over integers is less than `min_int` or greater than `max_int`) occurs during constraints internal computations and raise an assert failure only when using the byte code library `facile.cma`.

```
val (+~) : t -> t -> t
```

```
val (-~) : t -> t -> t
```

```
val ( *~) : t -> t -> t
```

Addition, substraction, multiplication over expressions.

```
val ( **~) : t -> int -> t
```

Exponentiation of an expression to an integer value.

```
val (/~) : t -> t -> t
```

```
val (%~) : t -> t -> t
```

Division and modulo over expressions. The exception `Division_by_zero` is raised whenever the second argument evaluates to 0.

```
val abs : t -> t
```

Absolute value over expressions.

```
val sum : t array -> t
```

```
val sum_fd : Var.Fd.t array -> t
```

sum exps (resp. sum_fd vars) returns the sum of all the elements of an array of expressions (resp. variables). Returns Fd.int 0 if the array is empty.

```
val scalprod : int array -> t array -> t
```

```
val scalprod_fd : int array -> Var.Fd.t array -> t
```

scalprod coeffs exps (resp. scalprod_fd coeffs vars) returns the scalar product of an array of integers and an array of expressions (resp. variables). Raises Invalid_argument if the arrays have not the same length.

```
val prod : t array -> t
```

```
val prod_fd : Var.Fd.t array -> t
```

prod exps (resp. prod_fd vars) returns the product of all the elements of an array of expressions (resp. variables).

Arithmetic Constraints over Expressions

```
val (<~)  : t -> t -> Cstr.t
```

```
val (<=~) : t -> t -> Cstr.t
```

```
val (=~)  : t -> t -> Cstr.t
```

```
val (>=~) : t -> t -> Cstr.t
```

```
val (>~)  : t -> t -> Cstr.t
```

```
val (<>~)  : t -> t -> Cstr.t
```

Constraints strictly less, less or equal, equal, greater or equal, strictly greater and different over expressions.

Reification

Reification of the arithmetic constraint which is replaced by an expression equal to a boolean variable instantiated to 1 if the constraint is satisfied and to 0 if it is violated. These operators are shortcuts to lighten the writing of reified expressions:

e1 op~~ e2 is equivalent to fd2e (Reify.boolean (e1 op~ e2)).

```
val (<~~)  : t -> t -> t
```

```
val (<=~~) : t -> t -> t
```

```
val (=~~)  : t -> t -> t
```

```
val (>=~~) : t -> t -> t
```

```
val (>~~)  : t -> t -> t
```

```
val (<>~~)  : t -> t -> t
```

Reified strictly less, less or equal, equal, greater or equal, strictly greater and different.

4.3 Module Cstr: Posting Constraints and Building New Ones

Overview

This module defines the type `t` of constraints and functions to create and post constraints: mainly a `create` function which allows to build new constraints from scratch (this function is not needed when using standard FaCiLe predefined constraints) and the mostly useful `post` function which must be called to effectively add a constraint to the constraint store.

Basic

```
exception DontKnow
```

Exception raised by the `check` function of a constraint (of type `t`) when it is not known whether the constraint is satisfied or violated.

```
type priority
```

Waking priority.

```
val immediate : priority
```

As soon as possible, for quick updates.

```
val normal : priority
```

Standard priority.

```
val later : priority
```

For time consuming constraints (e.g. `Gcc.cstr`, `Alldiff.cstr`, ...).

```
type t
```

The type of constraints.

```
val create : ?name:string -> ?fprint:(out_channel -> unit) -> ?priority:prio
rity -> ?init:(unit -> unit) -> ?check:(unit -> bool) -> ?not:(unit -> t) ->
 (unit -> bool) -> (t -> unit) -> t
```

`create ?name ?fprint ?priority ?init ?check ?not update delay` builds a new constraint.

- `name` is a describing string name of the constraint. Default value is `"anonymous"`.

- `fprint` (pretty-)printing of the constraint on an output channel taken as its only argument. Default value is to print the `name` string.

- `priority` is either `immediate`, `normal` or `later`. Time costly constraints should be waken after quick ones. Default value is `normal`.

- `init` is useful to perform initialization of auxiliary data structures needed and maintained by the `update` function. `init ()` is called as soon as the constraint is posted. Default value is `fun () -> ()`.

- `check` must be specified if the constraint is reifiable (as well as `not`). When the constraint is reified, `check ()` is called to verify whether the constraint is satisfied or violated, i.e. the constraint itself or its negation is entailed by the constraint store, and should return `true` if the constraint is satisfied, `false` if it is violated and raise

DontKnow when it is not known. check must not change the domains of the variables involved in the constraint. Default: Failure exception raised.

- not must be specified if the constraint is reifiable (as well as check). not () should return a constraint which is the negation of the constraint being defined. When the constraint is reified, it is called to post the negation of the constraint whenever check () return false, i.e. it is entailed by the constraint store. Default: Failure exception raised.

- update computes the domain reductions of the constraint. update () should return true when the constraint becomes solved, false if it is not yet entailed by the constraint store and raise Stak.Fail whenever a failure occurs. update is a mandatory argument.

- delay schedules the awakening of the constraint ct (which is taken as its only argument), i.e. the execution of its update function. If update () should be called (because it may reduce variables domains) when one of the events contained in the events list es occurred on variable v, then Var.delay es v ct should be called within the body of the delay function. delay is a mandatory argument.

```
val post : t -> unit
```

post c posts the constraint c to the constraint store.

```
val one : t
```

```
val zero : t
```

The constraint which succeeds (resp. fails) immediately.

Access

```
val id : t -> int
```

id c returns a unique integer identifying the constraint c.

```
val name : t -> string
```

name c returns the name of the constraint c.

```
val fprint : out_channel -> t -> unit
```

fprint chan c prints the constraint c on channel chan. Calls the fprint function passed to create.

```
val active_store : unit -> t list
```

active_store () returns the list of all active constraints, i.e. whose update functions have returned false.

4.4 Module Domain: Domain Operations

This module provides all necessary functions (and more) to create and handle domains, which are needed to create variables and perform propagation (i.e. domain reduction or filtering).

```
type t
```

Type of finite domains of integers (functional: no in place modifications, domains can be shared). Standard equality and comparison can be used on type domain.

Building New Domains

```
val empty : t
```

The empty domain.

```
val create : int list -> t
```

`create l` builds a new domain containing the values of `l`. Removes duplicates and sorts values. Returns `empty` if `l` is empty.

```
val unsafe_create : int list -> t
```

`unsafe_create l` builds a new domain containing the values of `l`. `l` must be sorted and must not contain duplicate values, otherwise behaviour is unspecified. Returns `empty` if `l` is empty. It is a more efficient variant of `create`.

```
val interval : int -> int -> t
```

`interval inf sup` returns the domain of all integers in the closed interval `[inf..sup]`. Raise `Invalid_argument` if `inf > sup`.

```
val int : t
```

The largest representable domain, handy to create variables for which bounds cannot be previously known. It is actually much smaller than `interval min_int max_int` (which really is the biggest one) to help to avoid overflows while computing bounds of expressions involving such variables.

```
val boolean : t
```

The domain containing `0` and `1`.

Access

```
val is_empty : t -> bool
```

`is_empty d` tests if the domain `d` is empty.

```
val size : t -> int
```

`size d` returns the number of integers in `d`.

```
val min : t -> int
```

```
val max : t -> int
```

`min d` (resp. `max d`) returns the lower (resp. upper) bound of `d`. If `d` is empty, the behaviour is unspecified (incorrect return value or exception raised).

```
val min_max : t -> int * int
```

`min_max d` returns both the lower and upper bound of `d`. If `d` is empty, the behaviour is unspecified (incorrect return value or exception raised).

```
val iter : (int -> unit) -> t -> unit
```

`iter f d` successively applies function `f` to all element of `d` in increasing order.

```
val interval_iter : (int -> int -> unit) -> t -> unit
```

`interval_iter f d` successively applies function `f` to the bounds of all the disjoint intervals of `d` in increasing order. E.g. a suitable function to print a domain can be `fun mini maxi -> Printf.printf "%d..%d " mini maxi`.

```
val member : int -> t -> bool
```

`member n d` tests if `n` belongs to `d`.

```
val values : t -> int list
```

`value d` returns the list of values of the domain `d`

```
val fprint : out_channel -> t -> unit
```

`fprint chan d` prints `d` on channel `chan`.

```
val sprint : t -> string
```

`sprint d` returns a string representation of `d`.

```
val included : t -> t -> bool
```

`included d1 d2` tests if domain `d1` is included in domain `d2`.

```
val smallest_geq : t -> int -> int
```

```
val greatest_leq : t -> int -> int
```

`smallest_geq dom val` (resp. `greatest_leq dom val`) returns the smallest (resp. greatest) value in `dom` greater (resp. smaller) than or equal to `val`. Raises `Not_found` if `max dom < val` (resp. `min dom > val`).

```
val choose : (int -> int -> bool) -> t -> int
```

`choose ord d` returns the mininum value of `d` for order `ord`. Raises `Not_found` if `d` is empty.

Operations

```
val remove : int -> t -> t
```

`remove n d` returns `d-n`. Returns `d` if `n` is not in `d`.

```
val add : int -> t -> t
```

`add n d` returns `d+n`.

```
val remove_up : int -> t -> t
```

```
val remove_low : int -> t -> t
```

`remove_up n d` (resp. `remove_low n d`) returns `d-[n+1..max_int]` (resp. `d-[min_int..n-1]`), i.e. removes values stricly greater (resp. less) than `n`.

```
val remove_closed_inter : int -> int -> t -> t
```

`remove_closed_inter inf sup d` returns `d-[inf..sup]`, i.e. removes values greater than or equal to `inf` and less or equal to `sup` in `d`. Returns `d` if `inf > sup`.

```
val intersection : t -> t -> t
```

```
val union : t -> t -> t
```

Intersection (resp. union) over domains.

```
val difference : t -> t -> t
```

`difference big small` returns `big-small`. `small` must be included in `big`, otherwise the behaviour is unspecified (incorrect return value or exception raised).

```
val remove_min : t -> t
```

```
val remove_max : t -> t
```

`remove_min d` (resp. `remove_max d`) returns `d` without its lower (resp. upper) bound. Raises `Invalid_argument` if `d` is empty.

```
val minus : t -> t
```

`minus d` returns the domain of opposite values of `d`.

```
val plus : t -> int -> t
```

`plus d n` translates a domain by `n`.

4.5 Module `FdArray`: Constraints over Arrays of Variables

```
val min : Var.Fd.t array -> Var.Fd.t
```

```
val max : Var.Fd.t array -> Var.Fd.t
```

`min vars` (resp. `max vars`) returns a variable constrained to be equal to the variable that will be instantiated to the minimal (resp. maximal) value among all the variables in the array `vars`. Raises `Invalid_argument` if `vars` is empty. Not reifiable.

```
val min_cstr : Var.Fd.t array -> Var.Fd.t -> Cstr.t
```

```
val max_cstr : Var.Fd.t array -> Var.Fd.t -> Cstr.t
```

`min_cstr vars mini` (resp. `max_cstr vars maxi`) returns the constraint `fd2e (min vars) =~ fd2e mini` (resp. `fd2e (max vars) =~ fd2e maxi`). Raises `Invalid_argument` if `vars` is empty. Not reifiable.

```
val get : Var.Fd.t array ->  Var.Fd.t -> Var.Fd.t
```

get vars index returns a variable constrained to be equal to vars.(index). Variable index is constrained within the range of the valid indices of the array (0..Array.length vars - 1). Raises Invalid_argument if vars is empty. Not reifiable.

```
val get_cstr : Var.Fd.t array -> Var.Fd.t -> Var.Fd.t -> Cstr.t
```

get_cstr vars index v returns the constraint fd2e vars.(index) =~ fd2e v. Variable index is constrained within the range of the valid indices of the array (0..Array.length vars - 1). Raises Invalid_argument if vars is empty. Not reifiable.

4.6 Module Gcc: Global Cardinality Constraint (a.k.a. Distribute)

```
type level = Basic | Medium | High
```

```
val cstr : ?level:level -> Var.Fd.t array -> (Var.Fd.t * int) array -> Cstr.
t
```

cstr (?level:High) vars distribution returns a constraint ensuring that for each pair (c,v) of cardinal variable c and integer value v in the list distribution, c variables in the array vars will be instantiated to v, i.e. card{vi = v | vi in vars} = c. All values v in distribution must be different otherwise the exception Invalid_argument is raised. Three levels of propagation are provided : Basic is the quickest, High performs the highest amount of propagation. level default value is High. The constraint posts the redundant constraint stating that the sum of the cardinals is equal to the number of variables. Not reifiable.

4.7 Module Goals: Building and Solving Goals

Overview

This module provides functions and operators to build goals that will control the search, i.e. mainly choose and instantiate variables.

```
type t
```

The type of goals.

Creation and Solving

```
val fail : t
```

```
val success : t
```

Failure (resp. success). Neutral element for the disjunction (resp. conjunction) over goals. Could be implemented as create (fun () -> Stak.fail "fail") (resp. create (fun () -> ())).

```
val atomic : ?name:string -> (unit -> unit) -> t
```

`atomic (?name:"atomic") f` returns a goal calling function `f`. `f` must take `()` as argument and return `()`. `name` default value is `"atomic"`.

```
val create : ?name:string -> ('a -> t) -> 'a -> t
```

`create (?name:"create") f a` returns a goal calling `f a`. `f` should return a goal (success to stop). `name` default value is `"create"`.

```
val create_rec : ?name:string -> (t -> t) -> t
```

`create_rec (?name:"create_rec") f` returns a goal calling `f`. `f` takes itself as argument and should return a goal (success to stop). Useful to write recursive goals. `name` default value is `"create_rec"`.

```
val solve : ?control:(int -> unit) -> t -> bool
```

`solve ?control g` solves the goal `g` and returns a success (`true`) or a failure (`false`). The execution can be precisely controlled with the `control` argument whose single argument is the number of bactracks since the beginning of the search. This function is called after every local failure:

- it can raise `Stak.Fail` to force a failure of the search in the current branch (i.e. backtrack);

- it can raise any other user exception to stop the search process;

- it must return `unit` to continue the search; this is the default behavior.

Operators

```
val (&&~) : t -> t -> t
```

```
val (||~) : t -> t -> t
```

Conjunction and disjunction over goals. Note that these two operators do have the SAME PRIORITY. Goals expressions must therefore be carefully parenthesized to produce the expected result.

```
val forto : int -> int -> (int -> t) -> t
```

```
val fordownto : int -> int -> (int -> t) -> t
```

`forto min max g` (resp. `fordownto min max g`) returns the conjunctive iteration of goal `g` on increasing (resp. decreasing) integers from `min` (resp. `max`) to `max` (resp. `min`).

```
val once : t -> t
```

`once g` cuts choice points left on goal `g`.

```
val sigma : ?domain:Domain.t -> (Var.Fd.t -> t) -> t
```

`sigma ?domain fgoal` creates the goal `(fgoal v)` where `v` is a new variable of domain `domain`. Default domain is the largest one. It can be considered as an existential quantification, hence the concrete notation `sigma` of this function (because existential quantification can be seen as a generalized disjunction).

Instantiation

```
val unify : Var.Fd.t -> int -> t
```

unify var x instantiates variable var to x.

```
val indomain : Var.Fd.t -> t
```

Non-deterministic instanciation of a variable, by labeling its domain (in increasing order).

```
val instantiate : (Domain.t -> int) -> Var.Fd.t -> t
```

instantiate choose var Non-deterministic instanciation of a variable, by labeling its domain using the value returned by the choose function.

```
val dichotomic : Var.Fd.t -> t
```

Non-deterministic instanciation of a variable, by dichotomic recursive exploration of its domain.

Goal Operations on Array of Variables

```
module Array :  sig
```

```
val foralli : ?select:('a array -> int) -> (int -> 'a -> t) -> 'a array -> t
```

foralli ?select g a returns the conjunctive iteration of goal g where order is computed by the heuristic ?select which must raise Not_found to terminate, i.e. g is successively applied to select a and a.(select a). Default heuristic is increasing order over indices.

```
val forall : ?select:('a array -> int) -> ('a -> t) -> 'a array -> t
```

forall ?select g a defined by foralli ?select (fun _i x -> g x) a.

```
val existsi : ?select:('a array -> int) -> (int -> 'a -> t) -> 'a array -> t
```

existsi ?select g a returns the disjunctive iteration of goal g where order is computed by the heuristic ?select which must raise Not_found to terminate, i.e. g is successively applied to select a and a.(select a). Default heuristic is increasing order over indices.

```
val exists : ?select:('a array -> int) -> ('a -> t) -> 'a array -> t
```

exists ?select g a defined by existsi ?select (fun _i x -> g x) a.

```
val choose_index : (Var.Attr.t -> Var.Attr.t -> bool) -> Var.Fd.t array -> i
nt
```

choose_index order fds returns the index of the best (minimun) free (not instantiated) variable in the array fds for the criterion order. Raises Not_found if all variables are bound (instantiated).

```
val not_instantiated_fd : Var.Fd.t array -> int
```

not_instantiated_fd fds returns the index of one element in fds which is not instantiated. Raises Not_found if all variables in array fds are bound.

```
val labeling: Var.Fd.t array -> t
```

Standard labeling defined by forall indomain.

```
end
```

Goal Operations on List of Variables

```
module List :  sig
```

```
    val forall : ?select:('a list -> 'a * 'a list) -> ('a -> t) -> 'a list -> t
```

forall ?select g [x1;x2;...;xn] is g x1 &&~ g x2 &&~ ... &&~ g xn, i.e. returns the conjunctive iteration of goal g on list a.

```
    val exists : ?select:('a list -> 'a * 'a list) -> ('a -> t) -> 'a list -> t
```

exists ?select g [x1;x2;...;xn] is g x1 ||~ g x2 ||~ ... ||~ g xn, i.e. returns the disjunctive iteration of goal g on list a.

```
    val member : Var.Fd.t -> int list -> t
```

member v l returns the disjunctive iteration of the instantiation of the variable v to the values in the integer list l. Defined by `fun v l -> exists (fun x -> create (fun () -> Fd.unify v x)) l`.

```
    val labeling: Var.Fd.t list -> t
```

Standard labeling defined by `forall indomain`.

```
end
```

Optimization

```
type bb_mode = Restart | Continue
```

Branch and bound mode.

```
val minimize : ?step:int -> ?mode:bb_mode -> t -> Var.Fd.t -> (int -> unit)
-> t
```

`minimize ?step ?mode goal cost solution` runs a Branch and Bound algorithm on `goal` for bound `cost`, with an improvement of at least `step` between each solution found. With `mode` equal to `Restart`, the search restarts from the beginning for every step whereas with mode `Continue` (default) the search simply carries on with an update of the cost constraint. `solution` is called with the instantiation value of `cost` (which must be instantiated by `goal`) as argument each time a solution is found; this function can therefore be used to store (e.g. in a reference) the current solution. Default `step` is 1. `minimize` always FAILS.

4.8 Module `Interval`: Variable Membership of an Interval

```
val is_member : Var.Fd.t -> int -> int -> Var.Fd.t
```

`is_member v inf sup` returns the boolean variable instantiated to `1` if `v` is in `inf..sup` and to `0` otherwise.

```
val cstr : Var.Fd.t -> int -> int -> Var.Fd.t -> Cstr.t
```

`cstr v inf sup b` returns the constraint ensuring that the boolean variable `b` is instantiated to `1` if `v` is in `inf..sup` and to `0` otherwise. Not reifiable.

4.9 Module Reify: Constraints Reification

All the reification functions and operators only work on REIFIABLE constraints, i.e. constraints which have a `check` function AND a `not` function (or built-in constraints for which the documentation does not mention "Not reifiable"). Otherwise a `Failure` ("fatal error") exception is raised.

```
val boolean : ?delay_on_negation:bool -> Cstr.t -> Var.Fd.t
```

`boolean (?delay_on_negation:true) c` returns a boolean (0..1) variable associated to the constraint `c`. The constraint is verified iff the boolean is equal to 1. The waking conditions of the contraint relating `c` and the boolean are the ones set by the `delay` function of `c` (set by the `delay` argument of `Cstr.create`). If the optional argument `delay_on_negation` is set to `true`, the new constraint is also attached to the events of the negation of `c` (i.e. the constraint returned by the `not` function of `c`), otherwise it is not. `delay_on_negation` default value is `true`.

```
val cstr : ?delay_on_negation:bool -> Cstr.t -> Var.Fd.t -> Cstr.t
```

`cstr delay_on_negation c b` is equivalent to the constraint `boolean ?delay_on_negation c =~ b`. `delay_on_negation` default value is `true`.

Operators

```
val (&&~~) : Cstr.t -> Cstr.t -> Cstr.t
```

```
val (||~~) : Cstr.t -> Cstr.t -> Cstr.t
```

```
val (=>~~) : Cstr.t -> Cstr.t -> Cstr.t
```

```
val (<=>~~) : Cstr.t -> Cstr.t -> Cstr.t
```

```
val xor : Cstr.t -> Cstr.t -> Cstr.t
```

```
val not : Cstr.t -> Cstr.t
```

Logical reification operators on constraints, namely and, or, implies, equivalent, exclusive or, not.

4.10 Module Sorting: Sorting Constraint

```
val sort : Var.Fd.t array -> Var.Fd.t array
```

`sort a` returns an array of variables constrained to be the variables in `a` sorted in increasing order.

```
val sortp : Var.Fd.t array -> Var.Fd.t array * Var.Fd.t array
```

`sortp a` same as `sort` but returns a couple `(sorted, perm)` where `sorted` is the array of sorted variables and `perm` is an array of variables constrained to be the permutation between `a` and `sorted`, i.e. `a.(i) = sorted.(perm.(i))`.

```
val cstr : Var.Fd.t array -> ?p:Var.Fd.t array option -> Var.Fd.t array -> C
str.t
```

`cstr a (?perm:None) sorted` returns the constraint ensuring that `sorted` is the result of sorting array `a` according to the permutation `perm`. `perm` default value is `None`, meaning the argument is irrelevant. Raises `Invalid_argument` if arrays have incompatible length. Not reifiable.

4.11 Module `Stak`: Global Stack of Goals, Backtrackable Operations

This module provides functions to control the execution of the goal stack, as well as backtrackable references. Nota: the module name `Stak` lacks a 'c' because of a possible clash with the Ocaml standard library module `Stack`.

```
type level
```

Type of a level in the stack.

```
val older : level -> level -> bool
```

`older l1 l2` true if level `l1` precedes `l2`.

```
exception Level_not_found of level
```

Raised by `cut` and `cut_bottom`.

```
val size : unit -> int
```

Size of the stack.

```
val level : unit -> level
```

Returns the current level.

```
val cut : level -> unit
```

`cut l` cuts the choice points left on the stack until level `l`. Raise `Level_not_found` if level is not reached and stack is empty.

```
type 'a ref
```

Backtrackable reference of type `'a`.

```
val ref : 'a -> 'a ref
```

Returns a reference which modification will be trailed during the solving of a goal.

```
val set : 'a ref -> 'a -> unit
```

Sets a reference.

```
val get : 'a ref -> 'a
```

Dereference.

```
val incr : int ref -> unit

val decr : int ref -> unit
```

Increments (resp. decrements) a reference.

Control of Failure

```
exception Fail of string
```

Raised during solving whenever a failure occurs. The string argument is informative.

```
val fail : string -> 'a
```

`fail x` equivalent to `raise (Fail x)`.

4.12 Module Var: Constrained, Attributed, Finite Domain Variables

The Attribute of a Domain Variable

```
module Attr :  sig
```

```
type t
```

Type of attributes.

```
val dom : t -> Domain.t
```

`dom a` returns the integer domain of an attribute.

```
type event
```

Type of events (modifications on variables) on which to suspend a constraint.

```
val on_refine : event
```

Event occuring when a variable is changed, i.e. its domain modified.

```
val on_subst : event
```

Event occuring when a variable is instantiated.

```
val on_min : event
```

```
val on_max : event
```

Event occuring when the lower (resp. upper) bound of a variable decreases.

```
val fprint : out_channel -> t -> unit
```

`fprint chan a` prints attribute `a` on channel `chan`.

```
val size : t -> int
```

`size a` returns the number of integer values in the domain associated with `a` (i.e. `dom a`).

```
val min : t -> int
```

```
val max : t -> int
```

`min a` (resp. `max a`) returns the lower (resp. upper) bound of `a`.

```
val values : t -> int list
```

`values a` returns the list of all integers in `dom a`.

```
val iter : (int -> unit) -> t -> unit
```

`iter f a` iterates f on each integer in `dom a`.

`val member : t -> int -> bool`

`member a n` tests if `n` belongs to `dom a`.

`val id : t -> int`

`id a` returns a unique integer identifying the attribute `a`.

`val constraints_number : t -> int`

`constraints_number a` returns the number of different constraints attached to `a`.

`end`

A Domain Variable

`type concrete_fd = Unk of Attr.t | Val of int`

Type of the value of a Fd variable.

`module Fd : sig`

Finite domain variable

`type t`

Type of Fd variable.

Creation

`val create : ?name:string -> Domain.t -> t`

`create ?name d` returns a new variable with domain `d`. If provided, `name` will be used by the pretty printer.

`val interval : ?name:string -> int -> int -> t`

`interval ?name inf sup` returns a new variable with domain `[inf..sup]`. If provided, `name` will be used by the pretty printer.

`val array : ?name:string -> int -> int -> int -> t array`

`array n inf sup` returns an array of `n` new variables with domain `[inf..sup]`. If provided, `name` (suffixed with the index of the element) will be used by the pretty printer.

`val int : int -> t`

`int n` returns a new variable instantiated to integer value `n`.

Access

`val is_var : t -> bool`

`is_var v` returns `true` if `v` is not instantiated and `false` otherwise.

`val value : t -> concrete_fd`

`value v` returns `Val n` if `v` is instantiated to `n`, `Unk a` otherwise where `a` is the attribute of `v`. Should always be used in a matching: `match value v with Val n -> ... | Unk a -> ...`.

```
val size : t -> int
```

size v returns the number of integer values in the domain of v (1 if v is instantiated.

```
val min : t -> int
```

min v returns the lower bound of v.

```
val max : t -> int
```

max v returns the upper bound of v.

```
val min_max : t -> int * int
```

min_max v returns both the lower and upper bounds of v.

```
val int_value : t -> int
```

int_value v returns the value of v if it is instantiated and raises a Failure exception otherwise.

```
val values : t -> int list
```

values v returns the list of all integers in the domain of v. If v is instantiated to n, returns the singleton list containing n.

```
val iter : (int -> unit) -> t -> unit
```

iter f v iterates f on each integer in the domain of v.

```
val member : t -> int -> bool
```

member v n returns true if n belongs to the domain of v and false otherwise.

```
val id : t -> int
```

id v returns a unique integer identifying the attribute associated with v. Must be called only on non ground variable, raise Failure otherwise.

```
val name : t -> string
```

name v returns then attributed to variable v (the empty string if it was not provided while created). Must be called only on non ground variable, raise Failure otherwise.

```
val compare : t -> t -> int
```

Compares two variables. Values (bound variables) are smaller than unknowns (unbound variables). Unknowns are sorted according to their attribute id.

```
val equal : t -> t -> bool
```

Tests if two variables are equal with respect to compare.

```
val fprint : out_channel -> t -> unit
```

fprint chan v prints variable v on channel chan.

```
val fprint_array : out_channel -> t array -> unit
```

fprint_array chan vs prints array of variables vs on channel chan.

Reduction

```
val unify : t -> int -> unit
```

`unify v n` instantiates variable `v` with integer value `n`. Raises `Stak.Fail` in case of failure. `unify` may be called either on unbound variables or on instantiated variables.

```
val refine : t -> Domain.t -> unit
```

`refine v d` reduces the domain of `v` with domain `d`. `d` must be included in the domain of `v`, otherwise the behaviour is unspecified (corrupted system or exception raised).

```
end

val delay : Attr.event list -> Fd.t -> Cstr.t -> unit
```

`delay event_list v c` suspends constraint `c` on all the events in `event_list` occurring on `v`. No effect on instantiated variables. To be used within new constraints "delay" functions.

4.13 Module `Easy`

`Easy` is a module that the user is strongly recommended to open to facilitate access to FaCiLe (unless names clash with other open modules). It simply defines aliases to values and types of other modules:

- All the infix operators from `Arith`, `Goals` and `Reify`
- Frequently used mapping functions of `Arith`: `i2e` and `fd2e`
- Type of finite domain variables from `Var`: `concrete_fd`
- Module `Fd` from `Var`

Note that the user of FaCiLe can extend this mechanism with its own "`Easier`" module aliasing any value or type of the library.

Index

Bibliography

[1] Nicolas Barnier and Pascal Brisset. FaCiLe: a functional constraint library. *ALP Newsletter*, 14(2), may 2001.

[2] Noelle Bleuzen Guernalec and Alain Colmerauer. Narrowing a $2n$-block of sorting in $O(n \log n)$. In *Principles and Practice of Constraint Programming*. Springer-Verlag, 1997.

[3] J. Hopcroft and R. Karp. An $n^{5/2}$ algorithm for maximum matching in bipartite graphs. *SIAM Journal of Computing*, 2(4):225–231, 1973.

[4] Serge Le Huitouze. A new data structure for implementing extensions to Prolog. In P. Deransart and J. Małuszyński, editors, *2nd Int. Work. Programming Languages Implementation and Logic Programming, LNCS 456*, pages 136–150. Springer-Verlag, 1990.

[5] Xavier Leroy. The Objective Caml System: User's and reference manual (`http://caml.inria.fr`), 2000.

[6] Jean-Charles Régin. Generalized arc consistency for global cardinality constraint. In *Proceedings of the Thirteenth National Conference on Artificial Intelligence*, 1996.